
SAS Code for Fitting Dynamic Groups Models 

In this supplement we provide SAS code and abridged output for the dynamic groups models 
for the two examples in the manuscript.  The first example concerns school effects when 
examining trajectories of growth in science achievement and uses data from the Longitudinal 
Study of American Youth (LSAY; Miller, Hoffer, Suchner, Brown, and Nelson, 1992).  The second 
example focuses on family effects within longitudinal data on developmental psychopathology 
and uses data from the Michigan Longitudinal Study (MLS; Zucker, Fitzgerald, Refior, Puttler, 
Pallas and Ellis, 2000). 

In fitting and interpreting models using the stabilizing banded structure, we make use of the 
companion macro file stableband.sas. 

Example 1: Schools as Dynamic Groups 

For this analysis the data set is referred to as canalysis and the variables are named and defined 
as follows: 

LSAYID a unique ID variable identifying the student 

schcode a unique ID variable identifying the school 

sci science achievement 

grade coded as 0 = 10th grade, 1 = 11th grade, 2=12th grade. 

cohort coded as 0 = first (began 10th grade in 1987), 1 = second (began 10th 
grade in 1990) 

year calendar year, represented by the last two digits (e.g., 1990 is coded as 
90) 

cstud_fund  school-mean-centered measure of a student’s fundamentalist attitudes 
towards science and religion (referenced in the manuscript as studentatt) 

schmean_fund school mean of students’ fundamentalist attitudes towards science and 
religion (referenced in the manuscript as schoolatt) 

cstud_ses  school-mean-centered measure of a student’s socioeconomic status 
(referenced in the manuscript as studentSES) 

schmean_ses school mean of students’ socioeconomic status (referenced in the 
manuscript as schoolSES) 

As described in the manuscript, we fit a sequence of models to this data varying in the 
covariance structure at the school level and in the inclusion of student- and school-level 
predictors.  We provide example code for all dynamic groups models; however, we present 
output only for the final, selected models. 

All unconditional models include the same fixed effects, as stipulated in the model statement; 
likewise, all conditional models include the same fixed effects (and include additional 
predictors).  Student-level trajectories are also allowed to differ in their intercepts and slopes 

(by grade) in all models, as indicated in the first random statement.  It is in the specification of 
the second random statement, for the school-level random effects, that the models differ.  
Random effects of year are included in each model, but the structure of these random effects 
varies between models as indicated in the type option.  It is important that the variable year is 
declared as a categorical variable within the class statement. 

Fitting Unconditional Models 

The following code specifies an unrestricted covariance structure for the school effects over 
time (note type=un in the second random statement): 

proc mixed data=canalysis method=reml maxiter=1000 cl; 

class lsayid schcode year;   

model sci=grade cohort grade*cohort/solution ddfm=bw notest alpha=.05; 

random intercept grade/subject=lsayid(schcode) type=un; 

random year / subject=schcode type=un;  

run; 

A Toeplitz structure can be specified for time-varying school effects as follows (type=toep): 

proc mixed data=canalysis method=reml maxiter=1000 cl; 

class lsayid schcode year;   

model sci=grade cohort grade*cohort/solution ddfm=bw notest alpha=.05; 

random intercept grade/subject=lsayid(schcode) type=un; 

random year / subject=schcode type=toep;  

run; 

The following code implements the CS structure (type=cs): 

proc mixed data=canalysis method=reml maxiter=1000 cl; 

class lsayid schcode year;   

model sci=grade cohort grade*cohort/solution ddfm=bw notest alpha=.05; 

random intercept grade/subject=lsayid(schcode) type=un; 

random year / subject=schcode type=cs;  

run; 

The AR(1) structure is obtained as follows  (type=ar(1)): 

proc mixed data=canalysis method=reml maxiter=1000 cl; 

class lsayid schcode year;   

model sci=grade cohort grade*cohort/solution ddfm=bw notest alpha=.05; 

random intercept grade/subject=lsayid(schcode) type=un; 

random year / subject=schcode type=ar(1);  

run; 

And the ARMA(1,1) structure is obtained as follows (type=arma(1,1)): 

proc mixed data=canalysis method=reml maxiter=1000; 

class lsayid schcode year;   

model sci=grade cohort grade*cohort/solution ddfm=bw notest alpha=.05; 

random intercept grade/subject=lsayid(schcode) type=un; 

random year / subject=schcode type=arma(1,1);  

run; 

The stabilizing banded structure is not a default structure available within SAS, so we make use 
of the flexibility of the type=lin(q) structure, or general linear covariance structure.  With this 
option, q is the number of parameters, and the structure is determined via a matrix constructed 
in a data step and input through the ldata option.  We have provided a macro, called 
stableband.sas, which automatically constructs this matrix within a data set named sb.  The 
stableband macro can be saved to a file and read in prior to fitting the model, as shown here: 

filename dyngrp 'C:\Users\bauer\documents\Projects\Dynamic Groups; 

%include dyngrp(SB.sas); 

(Replace the directory in the filename statement to the local directory in which stableband.sas 
has been saved).   

Alternatively, one can simply copy-paste the syntax within stableband.sas to run prior to the 
proc mixed syntax.  The syntax is given here: 

%macro stableband(lag=1,Gtimes=6); 

data sb; 

do p = 1 to &lag. + 1; 

 do i = 1 to &Gtimes.; 

   array col[&Gtimes.] col1-col&Gtimes.; 

   do j = 1 to &Gtimes.; 

      parm = p; 

   row = i; 

   if p < &lag. + 1 then do; 

     if (i-j) = p-1 then col[j] = 1; else col[j]=0; 

   end; 

   else do; 

  if j <= i - &lag. then col[j] = 1; else col[j]=0; 

   end; 

 end; 

    output;   

 end; 

end; 

drop j i p; 

run; 

%mend; 

Two arguments are required when calling the SB macro, specifically the number of time points 
the groups were observed (GTimes) and the lag at which the covariances stabilize (lag).  For 
instance, in the LSAY data there are up to six time points per school, so GTimes=6.  Our 
specification of the lag argument determines the covariance structure.  If we specify that the 
covariances stabilize at lag 1 then q=2 and we will replicate the type=cs structure.  At the other 
extreme, if we specify that the covariances stabilize at lag 5 then q=6 and we will replicate the 
type=toep structure.  Since we have already fit these models, we are more interested in 
situations in which the covariances stabilize at an intermediate lags, that is the covariance 
structures referenced in the manuscript as SB(2), SB(3), and SB(4). 

The SB(2) structure is fit with the following code: 

%stableband(lag=2,Gtimes=6); 

proc mixed data=canalysis method=reml maxiter=1000 cl; 

class lsayid schcode year;   

model sci=grade cohort grade*cohort/solution ddfm=bw notest alpha=.05; 

random intercept grade / subject=lsayid(schcode) type=un;  

random year / subject=schcode type=lin(3) ldata=sb;   

parms   

(100) 

(-1.5) 

(5) 

(12) 

(10) 

(10) 

(10) 

/lowerb=0,.,0,0,0,0,0;  

ods output covparms=covparms; 

run; 

Note that the call to the stableband macro precedes the proc mixed code, and that we 
specified lag=2.  We use type=lin(3) to indicate that we are using the general linear covariance 
structure with three parameters (the lag 1 and lag 2+ covariance parameters and the variance 
parameter), and we use ldata=sb to indicate the form of the covariance structure (recall that 
the stableband macro generated the sb data file). 

To fit the SB(3) structure, we simply modify the macro argument to lag=3 and indicate that now 
q=4 in the type=lin(q) option. 

%stableband(lag=3,Gtimes=6); 

proc mixed data=canalysis method=reml maxiter=1000 cl; 

class lsayid schcode year;   

model sci=grade cohort grade*cohort/solution ddfm=bw notest alpha=.05; 

random intercept grade / subject=lsayid(schcode) type=un;  

random year / subject=schcode type=lin(4) ldata=sb;   

parms   

(100) 

(-1.5) 

(5) 

(12) 

(10) 

(10) 

(10) 

(10) 

/lowerb=0,.,0,0,0,0,0,0;  

run; 

Likewise, the SB(4) structure is fit with the following code: 

%stableband(lag=4,Gtimes=6); 

proc mixed data=canalysis method=reml maxiter=1000 cl; 

class lsayid schcode year;   

model sci=grade cohort grade*cohort/solution ddfm=bw notest alpha=.05; 

random intercept grade / subject=lsayid(schcode) type=un;  

random year / subject=schcode type=lin(5) ldata=sb;   

parms   

(100) 

(-1.5) 

(5) 

(12) 

(10) 

(10) 

(10) 

(10) 

(10) 

/lowerb=0,.,0,0,0,0,0,0,0;  

run; 

In addition to the syntax options already described, the code for the SB structures makes use of 
the parms statement to specify starting values for the variance and covariance parameters.  
The addition of this syntax is essential to prevent SAS from using start values that would 
produce a non-invertible school-level covariance matrix.  The start values (within parentheses) 
are in the order of the covariance parameters.   

 The first three entries are start values for the variance of the student-level intercepts 
(100), covariance of student-level intercepts and slopes (-1.5), and variance of the 
student-level slopes (5).   

 The next entry is the start value for the variance and of the school-level effects (12).   

 The following q-1 entries correspond to the lagged covariances (which are all set to a 
start value of 10 here; it is important that this value be less than the variance, e.g., 
10<12).   

 Finally, the last entry is the start value for the time-specific residual variance (also 
started at 10).  

The specific start values used here were loosely based on the results obtained from the type=cs 
model.  Confirmation that the variance and covariance parameter start values have been 
ordered correctly can be obtained from the Covariance Parameter Estimates table produced by 
proc mixed upon running the model.  The order of the start values should correspond to the 
order of the estimates shown in this table. 

Within the parms statement we have also specified a lower bound of zero for all parameters 
other than the covariance of the student-level intercepts and slopes.  This enforces positive 
variances and also positive covariances for the school effects.  (Negative over-time covariances 
among school effects are unlikely but not impossible.  For situations in which negative 
covariances are more plausible, these lower bounds could be removed). 



6 

 
Fitting Conditional Models 

Conditional models were fit using syntax that was nearly identical to the syntax for the 
unconditional models, with the exception that additional fixed effects were included for the 
predictors in the model statement.  That is, the model statement was modified to be 

model sci = grade cohort grade*cohort  

            cstud_fund cstud_ses  

            schmean_fund schmean_ses/  

            solution ddfm=bw notest alpha=.05; 

Output from Optimally Fitting Models 

As described in the manuscript, the optimally fitting covariance structure for both the 
unconditional and conditional models was the SB(4) structure.  Abridged results from the 
unconditional model are shown here: 
                                      The Mixed Procedure 

 

                                       Model Information 

 

                     Data Set                     WORK.CANALYSIS 

                     Dependent Variable           sci 

                     Covariance Structures        Unstructured, Linear, 

                                                  Variance Components 

                     Subject Effects              LSAYID(schcode), schcode 

                     Estimation Method            REML 

                     Residual Variance Method     Parameter 

                     Fixed Effects SE Method      Model-Based 

                     Degrees of Freedom Method    Between-Within 

 

 

                                          Dimensions 

 

                              Covariance Parameters             9 

                              Columns in X                      4 

                              Columns in Z Per Subject        304 

                              Subjects                         51 

                              Max Obs Per Subject             399 

 

 

                                    Number of Observations 

 

                          Number of Observations Read            7756 

                          Number of Observations Used            7756 

                          Number of Observations Not Used           0 

 

 

                                       Iteration History 

 

                  Iteration    Evaluations    -2 Res Log Like       Criterion 

 

                          1              2     52134.85855691      0.00155910 

                          2              1     52121.72942851      0.00039101 

                          3              1     52112.86374848      0.00009613 

                          4              1     52110.65980331      0.00002071 

                          5              1     52110.20392529      0.00000200 

                          6              1     52110.16348783      0.00000002 

                          7              1     52110.16302501      0.00000000 

 

 

                                   Convergence criteria met. 

 

 

                                 Covariance Parameter Estimates 

 

           Cov Parm     Subject            Estimate     Alpha       Lower       Upper 

 

           UN(1,1)      LSAYID(schcode)     93.7106      0.05     88.9250     98.8951 

           UN(2,1)      LSAYID(schcode)     -1.2697      0.05     -2.7480      0.2087 

           UN(2,2)      LSAYID(schcode)      4.7079      0.05      4.0213      5.5878 

           LIN(1)       schcode             14.5683      0.05     10.0678     22.9538 

           LIN(2)       schcode             13.5601      0.05      9.1585     22.1314 

           LIN(3)       schcode             12.8337      0.05      8.5128     21.5499 

           LIN(4)       schcode             12.2778      0.05      8.0063     21.1907 

           LIN(5)       schcode             10.7946      0.05      6.6717     20.3949 

           Residual                         10.1192      0.05      9.4815     10.8238 

 

 

                                        Fit Statistics 

 

                             -2 Res Log Likelihood         52110.2 

                             AIC (smaller is better)       52128.2 

                             AICC (smaller is better)      52128.2 

                             BIC (smaller is better)       52145.5 

 

 

                                  Solution for Fixed Effects 

 

                           Standard 

 Effect         Estimate      Error     DF   t Value   Pr > |t|    Alpha      Lower      Upper 

 

 Intercept       60.4817     0.5806     50    104.18     <.0001     0.05    59.3157    61.6478 

 grade            2.4860     0.1580   7702     15.73     <.0001     0.05     2.1762     2.7957 

 COHORT           1.4808     0.4795   7702      3.09     0.0020     0.05     0.5409     2.4207 

 grade*COHORT    -0.6219     0.2389   7702     -2.60     0.0092     0.05    -1.0901    -0.1536 

These estimates match the results shown in Table 3 (Model 1) of the manuscript.  See the 
manuscript for interpretation of the fixed effects. 

The student-level covariance parameter estimates are interpreted as follows: 

 UN(1,1) is the variance of the student-level trajectory intercepts (variability in 10th grade 
science achievement) 

 UN(2,1) is the covariance of the student-level intercepts and slopes 

 UN(2,2) is the variance of the student-level trajectory slopes (variability in change over 
time) 

The school-level covariance parameter estimates are interpreted as follows: 

 The LIN(1) parameter estimate corresponds to the variance of the school effects. 

 The LIN(2) parameter estimate corresponds to the lag 1 covariance, the LIN(3) 
parameter estimate corresponds to the lag 2 covariance, and the LIN(4) parameter 
estimate corresponds to the lag 3 covariance. 

 The LIN(5) parameter estimate corresponds to the lag 4+ covariance (lag 4 and lag 5). 

These values can also be used to construct the model-implied covariance and correlation 
matrices of the school effects.  These matrices can be produced using proc iml within SAS.  To 
do so, the first thing we need to do is output the values of the covariance parameter estimates 
produced by proc mixed.  To do this, we re-run the model using an ods output statement, as 
follows: 

%stableband(lag=4,Gtimes=6); 

proc mixed data=canalysis method=reml maxiter=1000 cl; 

class lsayid schcode year;   

model sci=grade cohort grade*cohort/solution ddfm=bw notest alpha=.05; 

random intercept grade / subject=lsayid(schcode) type=un gcorr;  

random year / subject=schcode type=lin(5) ldata=sb;   

parms   

(100) 

(-1.5) 

(5) 

(12) 

(10) 

(10) 

(10) 

(10) 

(10) 

/lowerb=0,.,0,0,0,0,0,0,0;  

ods output covparms=covparms; 

run; 

The ods output statement puts the “Covariance Parameter Estimates” table of output, 
referenced internally by SAS as covparms, into a data set also called covparms for us to access 
within proc iml.   

proc iml; 

 use covparms; 

 read all var{Estimate} where(subject="schcode") into Linq ; 

 print Linq; 

 Gtimes=6; stablelag=4; 

 cov = J(Gtimes,Gtimes,0); 

 do i = 1 to Gtimes; 

   do j = 1 to Gtimes; 

      lag = abs(i-j); 

   if lag < NROW(linq) then do; 

         cov[i,j] = linq[lag+1]; 

   end; 

   else do; 

     cov[i,j] = linq[stablelag+1]; 

   end; 

   end; 

 end; 

 print cov; 

 corr = inv(sqrt(diag(cov)))*cov*inv(sqrt(diag(cov))); 

 print corr; 

quit; 

The snippet “where(subject="schcode")” selects only the school-level covariance parameter 
estimates.  For other applications this code should be modified to reference the appropriate 
group-level ID variable.  Additionally, the code “Gtimes=6; stablelag=4;” is used to define the 
number of time points over which the groups were observed and to define the lag at which the 
covariances stabilize (here lag 4).  These values too should be modified to be appropriate to the 
specific application. 

The output produced by proc iml is shown here: 
                                              Linq 

 

                                           14.568339 

                                           13.560148 

                                           12.833693 

                                           12.277768 

                                           10.794648 

 

 

                                              cov 

 

                  14.568339 13.560148 12.833693 12.277768 10.794648 10.794648 

                  13.560148 14.568339 13.560148 12.833693 12.277768 10.794648 

                  12.833693 13.560148 14.568339 13.560148 12.833693 12.277768 

                  12.277768 12.833693 13.560148 14.568339 13.560148 12.833693 

                  10.794648 12.277768 12.833693 13.560148 14.568339 13.560148 

                  10.794648 10.794648 12.277768 12.833693 13.560148 14.568339 

 

 

                                              corr 

 

                          1 0.9307957 0.8809304 0.8427706 0.7409663 0.7409663 

                  0.9307957         1 0.9307957 0.8809304 0.8427706 0.7409663 

                  0.8809304 0.9307957         1 0.9307957 0.8809304 0.8427706 

                  0.8427706 0.8809304 0.9307957         1 0.9307957 0.8809304 

                  0.7409663 0.8427706 0.8809304 0.9307957         1 0.9307957 

                  0.7409663 0.7409663 0.8427706 0.8809304 0.9307957         1 

The column of values labeled “Linq” repeats the covariance parameter estimates at the school 
level.  The output labeled “cov” and “corr” corresponds to the covariance and correlation 
matrices of the school effects, respectively.  Thus we see that the correlation in the school 
effects between adjacent years is .93.  Across two years the correlation drops to .88.  Across 
three years, the correlation drops to .84.  The correlation stabilizes at four or more years of 
separation at .74.  Plotting these correlations produces the trend shown in Figure 2 (Model 1) of 
the manuscript. 

The optimally fitting conditional model also used the SB(4) covariance structure for the school 
effects.  Output for the covariance parameter estimates and fixed effects estimates match the 
values shown in Table 3 (Model 2): 

 
                                 Covariance Parameter Estimates 

 

           Cov Parm     Subject            Estimate     Alpha       Lower       Upper 

 

           UN(1,1)      LSAYID(schcode)     87.9728      0.05     83.4454     92.8807 

           UN(2,1)      LSAYID(schcode)     -1.4188      0.05     -2.8575     0.02002 

           UN(2,2)      LSAYID(schcode)      4.7040      0.05      4.0180      5.5831 

           LIN(1)       schcode              9.9908      0.05      6.8100     16.0757 

           LIN(2)       schcode              8.9921      0.05      5.9212     15.2765 

           LIN(3)       schcode              8.1997      0.05      5.2306     14.6766 

           LIN(4)       schcode              7.6346      0.05      4.7267     14.3813 

           LIN(5)       schcode              6.1826      0.05      3.4611     14.0573 

           Residual                         10.1192      0.05      9.4817     10.8235 

 

 

                                        Fit Statistics 

 

                             -2 Res Log Likelihood         51883.8 

                             AIC (smaller is better)       51901.8 

                             AICC (smaller is better)      51901.9 

                             BIC (smaller is better)       51919.2 

 

 

                                  Solution for Fixed Effects 

 

                           Standard 

 Effect         Estimate      Error     DF   t Value   Pr > |t|    Alpha      Lower      Upper 

 

 Intercept       60.5443     0.4953     48    122.25     <.0001     0.05    59.5485    61.5401 

 grade            2.4888     0.1603   7700     15.53     <.0001     0.05     2.1747     2.8030 

 COHORT           1.3183     0.4743   7700      2.78     0.0055     0.05     0.3885     2.2480 

 grade*COHORT    -0.6105     0.2440   7700     -2.50     0.0124     0.05    -1.0888    -0.1322 

 cstud_fund      -2.6751     0.3530   7700     -7.58     <.0001     0.05    -3.3671    -1.9831 

 cstud_ses        0.1253    0.01116   7700     11.23     <.0001     0.05     0.1034     0.1472 

 schmean_fund    -8.3805     3.9987     48     -2.10     0.0414     0.05   -16.4204    -0.3407 

 schmean_ses      0.1211     0.1057     48      1.15     0.2578     0.05   -0.09152     0.3337 

The fixed effects are interpreted within the manuscript.  The school-level covariance 
parameters are labeled as described for the unconditional model above.  These parameters are 
also interpreted similarly – except that they now represent the residual (co)variances.  As 
illustrated above, these estimates can be arrayed into a residual covariance matrix and used to 
construct a residual correlation matrix.  These correlations can be plotted to produce the trend 
shown in Figure 2 (Model 2). 

Example 2:  Families as Dynamic Groups 

For this analysis the data set is referred to as allkidssubset and the variables are named and 
defined as follows: 

kidid a unique ID variable identifying the child 

family a unique ID variable identifying the school 

intscore depression score for the child 

extscore externalizing behavior score for the child 

ageyrc age of the child in years, centered at age 14 

intyear calendar year when interview assessing child depression and 
externalizing behavior were conducted 

kidgen  sex of the child, coded 0=female, 1=male  

coa indicator of parental history of alcohol disorder; coded 0=no, 1=yes 

parentanti  indicator of parental history of antisocial personality disorder; coded 
0=no, 1=yes 

parentdep indicator of parental history of depression/dysthymia; coded 0=no, 1=yes 

 

To reduce redundancy with the documentation of the prior example, here we focus specifically 
on the code for the optimally fitting dynamic groups models.  For this example, an AR(1) 
structure for time-varying family effects resulted in the best fit to the data.  The unconditional 
model (model 1) was fit via the following code: 

proc mixed data=allkidssubset method=reml covtest cl maxiter=1000; 

class family intyear kidid; 

model extscore=ageyrc ageyrc*ageyrc/solution ddfm=bw notest alpha=.05; 

random intercept ageyrc ageyrc*ageyrc/subject=kidid(family) type=un; 

random intyear/subject=family type=ar(1); 

run; 

 

proc mixed data=allkidssubset method=reml covtest maxiter=1000; 

class family intyear kidid; 

model intscore=ageyrc ageyrc*ageyrc/solution ddfm=bw notest alpha=.05; 

random intercept ageyrc ageyrc*ageyrc/subject=kidid(family) type=un; 

random intyear/subject=family type=ar(1); 

run; 

Note the type=ar(1) specification in the second random statement for each outcome. 

The conditional model (model 2) differed in the inclusion of additional predictors, augmenting 
the model statement for each outcome: 

proc mixed data=allkidssubset method=reml covtest cl maxiter=1000; 

class family intyear kidid; 

model extscore=ageyrc ageyrc*ageyrc kidgen kidgen*ageyrc  

      kidgen*ageyrc*ageyrc coa parentanti parentdep/solution ddfm=bw  

      notest alpha=.05; 

random intercept ageyrc ageyrc*ageyrc/subject=kidid(family) type=un; 

random intyear/subject=family type=ar(1); 

run; 

 

proc mixed data=allkidssubset method=reml covtest cl maxiter=1000; 

class family intyear kidid; 

model intscore=ageyrc ageyrc*ageyrc kidgen kidgen*ageyrc coa  

      parentanti parentdep/solution ddfm=bw notest alpha=.05; 

random intercept ageyrc ageyrc*ageyrc/subject=kidid(family) type=un; 

random intyear/subject=family type=ar(1); 

run; 

Below we show sample output from the unconditional model for externalizing behavior: 
                                 Covariance Parameter Estimates 

 

                                       Standard       Z 

   Cov Parm   Subject        Estimate     Error   Value      Pr Z   Alpha     Lower     Upper 

 

   UN(1,1)    kidID(family)    0.2806   0.03452    8.13    <.0001    0.05    0.2235    0.3627 

   UN(2,1)    kidID(family)   0.01289  0.005836    2.21    0.0272    0.05  0.001451   0.02433 

   UN(2,2)    kidID(family)  0.007864  0.002226    3.53    0.0002    0.05  0.004835   0.01499 

   UN(3,1)    kidID(family)  -0.00586  0.002918   -2.01    0.0446    0.05  -0.01158  -0.00014 

   UN(3,2)    kidID(family)  -0.00154  0.000633   -2.44    0.0149    0.05  -0.00278  -0.00030 

   UN(3,3)    kidID(family)  0.000746  0.000436    1.71    0.0437    0.05  0.000307  0.003723 

   Variance   family           0.1358   0.02676    5.07    <.0001    0.05   0.09552    0.2084 

   AR(1)      family           0.8190   0.04881   16.78    <.0001    0.05    0.7233    0.9147 

   Residual                    0.2098   0.01059   19.81    <.0001    0.05    0.1905    0.2322 

 

 

                                        Fit Statistics 

 

                             -2 Res Log Likelihood          4825.9 

                             AIC (smaller is better)        4843.9 

                             AICC (smaller is better)       4844.0 

                             BIC (smaller is better)        4876.6 

 

 

 

                                   Solution for Fixed Effects 

 

                            Standard 

 Effect          Estimate      Error     DF   t Value   Pr > |t|    Alpha      Lower      Upper 

 

 Intercept         0.3899    0.03339    278     11.68     <.0001     0.05     0.3242     0.4556 

 ageyrc          0.008677   0.008429   2182      1.03     0.3034     0.05   -0.00785    0.02521 

 ageyrc*ageyrc   -0.01488   0.003625   2182     -4.10     <.0001     0.05   -0.02199   -0.00777 

Focusing on the covariance parameter estimates, the child-level parameters describe variability 
in the growth trajectories of externalizing behavior with age: 

 UN(1,1) is the variance of the child-level trajectory intercepts (variability in externalizing 
behavior at age 14) 

 UN(2,1) is the covariance of the child-level intercepts and linear slopes 

 UN(2,2) is the variance of the child-level trajectory linear slopes (rates of change in 
externalizing behavior at age 14) 

 UN(3,1) is the covariance of the child-level intercepts and quadratic slopes 

 UN(3,2) is the covariance of the child-level linear and quadratic slopes 

 UN(3,3) is the variance of the child-level trajectory quadratic slopes (rates of 
acceleration/deceleration in change over time) 

The family-level covariance parameter estimates are interpreted as follows: 

 The Variance parameter estimate corresponds to the variance of the family effects. 

 AR(1) parameter estimate corresponds to the autocorrelation of the family effects. 

The AR(1) estimate of .819 can be interpreted as the correlation of family effects across a one-
year interval. Across a n-year interval, the correlation is implied to be .819n, and one can easily 
compute these correlations across all of the observed intervals to produce a plot of the 
correlations over time. 

Sample output for the conditional model for externalizing behavior is shown here: 
                                 Covariance Parameter Estimates 

 

                                       Standard       Z 

   Cov Parm   Subject        Estimate     Error   Value      Pr Z   Alpha     Lower     Upper 

 

   UN(1,1)    kidID(family)    0.2646   0.03094    8.55    <.0001    0.05    0.2130    0.3375 

   UN(2,1)    kidID(family)   0.01617  0.005313    3.04    0.0023    0.05  0.005758   0.02659 

   UN(2,2)    kidID(family)  0.007892  0.002114    3.73    <.0001    0.05  0.004966   0.01446 

   UN(3,1)    kidID(family)  -0.00725  0.002751   -2.64    0.0084    0.05  -0.01264  -0.00186 

   UN(3,2)    kidID(family)  -0.00107  0.000595   -1.79    0.0730    0.05  -0.00223  0.000100 

   UN(3,3)    kidID(family)  0.000651  0.000417    1.56    0.0595    0.05  0.000251  0.004060 

   Variance   family          0.08608   0.02154    4.00    <.0001    0.05   0.05565    0.1507 

   AR(1)      family           0.7568   0.07341   10.31    <.0001    0.05    0.6129    0.9006 

   Residual                    0.2124   0.01069   19.87    <.0001    0.05    0.1929    0.2350 

 

                                        Fit Statistics 

 

                             -2 Res Log Likelihood          4736.5 

                             AIC (smaller is better)        4754.5 

                             AICC (smaller is better)       4754.5 

                             BIC (smaller is better)        4787.2 

 

 

                                  Solution for Fixed Effects 

 

                                 Standard 

 Effect                Estimate     Error    DF  t Value  Pr > |t|   Alpha     Lower     Upper 

 

 Intercept              -0.1329   0.07039   275    -1.89    0.0601    0.05   -0.2715  0.005663 

 ageyrc                 0.05514   0.01459  2179     3.78    0.0002    0.05   0.02653   0.08376 

 ageyrc*ageyrc         -0.03425  0.006716  2179    -5.10    <.0001    0.05  -0.04742  -0.02108 

 kidgen                  0.2168   0.06173  2179     3.51    0.0005    0.05   0.09573    0.3378 

 ageyrc*kidgen         -0.07006   0.01723  2179    -4.07    <.0001    0.05   -0.1038  -0.03628 

 ageyrc*ageyrc*kidgen   0.02772  0.008011  2179     3.46    0.0005    0.05   0.01201   0.04343 

 coa                     0.4153   0.06317   275     6.57    <.0001    0.05    0.2909    0.5396 

 parentanti              0.2065   0.07976   275     2.59    0.0101    0.05   0.04948    0.3635 

 parentdep              0.09789   0.06362   275     1.54    0.1251    0.05  -0.02736    0.2231 

The covariance parameter estimates are labeled and interpreted as indicated above, with the 
exception that these are now residuals. 

See the primary manuscript for further interpretation of these results. 

